

```

10 REM*****
20 REM*****
30 REM*****
40 REM*****      Loesungsvorschlag      *****
50 REM*****      Filtermessung          *****
60 REM*****      Uebung c)+d)           *****
70 REM*****
80 REM*****      GST 11/88              *****
90 REM*****
100 REM
110 REM ##### HAUPTPROGRAMM - BEGINN #####
120 REM
130 GOSUB Pca_init: REM 1000
140 GOSUB Var_init: REM 1100
150 GOSUB Graf_var: REM 1200
160 GOSUB Ieee_init: REM 1400
170 GOSUB Dmm_init: REM 1500
180 GOSUB Gen_init: REM 1600
185 GOSUB Referenzwerteinstellung: REM 2200
190 GOSUB Messung: REM 2000
200 GOSUB Datei: REM 3000
210 REM
220 REM
998 END
999 REM ##### HAUPTPROGRAMM - ENDE #####
1000Pca_init: REM+++++
1010 REM
1020 PRINT "E-[2J": CLEAR
1030 REM
1040 REM
1099 RETURN : REM+++++
1100Var_init: REM+++++
1110 REM
1120 Untere_frequenz=430: Obere_frequenz=480
1130 Unterer_pegel=-80: Oberer_pegel=0
1140 Schrittweite=0.5
1150 DIM Speicher_freq(100),Speicher_pegel(100)
1160 Speicherzaehler=0
1170 Terminator=10: Time_out=2000: REM *** iec-bus-variable ***
1180 REM
1190 REM
1199 RETURN : REM+++++
1200Graf_var: REM+++++
1210 REM
1220 REM
1230 Xl=Untere_frequenz: Xh=Obere_frequenz
1240 Yl=Unterer_pegel: Yh=Oberer_pegel
1250 Lx$="Frequenz in kHz": Ly$="Pegel/dB"
1260 L$="Filtermessung"
1270 Lg=0: Co=0: REM *** Ausgabe Lin/Lin und schwarz/weiss ***
1280 Sf=0: REM *** Keine Sonderfunktion ***
1290 GOSUB Graf_init: REM 25000 --> TEIL VON GRAFIKROUTINE GPH-PCA
1299 RETURN : REM+++++
1300Graf_akt: REM+++++
1310 REM
1320 REM
1330 REM
1340 Xw=Frequenz
1350 Yw=Messwert
1360 GOSUB Graf_exe: REM 25085 --> TEIL VON GRAFIKROUTINE GPH-PCA
1370 REM
1380 REM
1390 RETURN : REM+++++

```

```

1400Ieee_init: REM+++++
1410 REM
1430 IEC IFC ,DCL ,TERM Terminator,TIME Time_out
1440 REM
1450 REM
1499 RETURN : REM+++++
1500Dmm_init: REM+++++
1510 REM
1520 REM
1530 Ure=7
1540 Gfdmm$="U5,F2,N1,WD,X3"
1550 GOSUB Dmm_out: REM 1700
1560 REM
1570 REM
1599 RETURN : REM+++++
1600Gen_init: REM+++++
1610 REM
1620 REM
1630 Spn=11: REM
1640 Gfgen$="R5,400KH,2V": REM
1650 GOSUB Gen_out: REM 1750
1660 REM
1670 REM
1699 RETURN : REM+++++
1700Dmm_out: REM+++++
1710 REM
1720 IEC OUT Ure,Gfdmm$
1730 REM
1740 RETURN : REM+++++
1750Gen_out: REM+++++
1760 REM
1770 IEC OUT Spn,Gfgen$
1780 REM
1790 REM
1799 RETURN : REM+++++
1800Dmm_in: REM+++++
1810 REM
1820 IEC IN Ure,Messwert$
1830 REM
1840 REM
1899 RETURN : REM+++++
2000Messung:
2010 REM
2020 FOR Frequenz=Untere_frequenz TO Obere_frequenz STEP Schrittweite
2030     Gfgen$=STR$(Frequenz)+"KH"
2040     GOSUB Gen_out: REM 1750
2050     GOSUB Dmm_in: REM 1800
2060     GOSUB Messwertbearbeitung: REM 2100
2070     GOSUB Graf_akt: REM1300
2080 NEXT Frequenz
2090 REM
2099 RETURN : REM+++++
2100Messwertbearbeitung: REM+++++
2110 REM
2120 REM
2130 Messwert=VAL(Messwert$)
2140 Speicher_freq(Speicherzaehler)=Frequenz
2150 Speicher_pegel(Speicherzaehler)=Messwert
2160 Speicherzaehler=Speicherzaehler+1
2170 REM
2180 REM
2199 RETURN : REM+++++

```

HIER KOMMT EIN ANDERER GENERATOR  
ZUR ANWENDUNG ! VERGLEICHEN SIE  
DIE MINIMALEN AENDERUNGEN !

```

2200ReferenzwertEinstellung: REM+++++
2210 REM
2220 Gfgen$="455KH": REM Generator auf Mittenfrequenz (Durchlassmaximum)
2230 GOSUB Gen_out
2240 Gfdmm$="X2": REM Triggerung und Uebernahme als Referenzwert im URE
2250 GOSUB Dmm_out: HOLD 2000
2260 Gfdmm$="X3": REM wieder urspruengliche Triggerung (bei Talkeraufruf)
2270 GOSUB Dmm_out
2280 REM
2299 RETURN : REM+++++
3000Datei: REM+++++
3010 REM
3020 OPEN# 1,"FILTER.DAT"
3025 PRINT# 1,"----- FREQUENZ ----- PEGEL ----- "
3030 REPEAT
3040   PRINT# 1,USING "#####.##";Speicher_freq(I);" KHZ ";TAB(15);
3050   PRINT# 1,USING "##.###";Speicher_pegel(I);" dB "
3055   I=I+1
3060 UNTIL I>100
3070 HOLD 1000
3080 CLOSE# 1
3090 CLEAR : SHELL "TYPE FILTER.DATIMORE"
3099 RETURN : REM+++++
3100 REM
3110 REM
3120 REM
3130 REM   Ab hier ist die Aufgabe beendet. Von Zeile 20000 an folgt die
3140 REM   Standardgrafikroutine GPH-PCA !
3150 REM
3160 REM
3170 REM
4000 REM *****
4010 REM *****
4020 REM
4030 REM In Uebung c) sind gegenueber der Uebung b) nur folgende Punkte abzu-
4040 REM aendern :
4050 REM
4060 REM -Einfuegen eines Unterprogramms, das bei der Mittenfrequenz 455 kHz
4070 REM   den gemessenen Wert als Referenzwert im URE uebernimmt, die fol-
4080 REM   genden Messungen intern umrechnet und die Messwerte als dB- Ab-
4090 REM   weichungen zur Verfuegung stellt.(Zeile 2200 - 2299, sowie Aufruf
4100 REM   dieses Unterprogramms in Zeile 185)
4110 REM
4120 REM -Aenderung der Bildschirmausgaben V ---> dB in den Zeilen 1250 und
4130 REM   3050
4140 REM
4150 REM -Aenderung der Initialisierungswerte fuer den Grafikrahmen in Zeile
4160 REM   1130 (Der obere Pegel erhaelt den Wert 0 dB, der untere Wert wird
4170 REM   wegen der Darstellung einer Daempfung negativ <hier -80 dB>).
4180 REM
4190 REM -Schliesslich muss das URE mit dem Geraetebefehl "U5" in Zeile 1540
4200 REM   auf Berechnung und Ausgabe der "Delta"-dB Werte umgeschaltet werden
4210 REM
4215 REM -Ausserdem wird ein anderes Generatormodell verwendet
4220 REM ~~~~~
4230 REM
4240 REM In Uebung d) werden nur Frequenzgrenzen, Schrittweite und evtl. die
4250 REM Dimensionierungen fuer Speicherfrequenz und -pegel abgeaendert.
4260 REM Vernuenftige Werte dafuer sollen Sie aus der vorherigen Messung in
4270 REM Teil c) entnehmen.
4280 REM
4290 REM Natuerlich muss auch der Grafikrahmen mit seinen Variablen in
4300 REM Graf_var entsprechend angepasst werden.
4310 REM
4320 REM
4330 REM

```